

How to Model Mediating and Moderating Effects

Hsueh-Sheng Wu
CFDR Workshop Series
February 22, 2021

BGSU

 Center for Family and Demographic Research

Outline


- What are moderator and mediator?
- How to estimate mediation effects?
- How to estimate moderation effects?
- Stata codes
- Conclusions

What Are Moderator and Mediator?


1. A bivariate relation between X and Y



2.1. Z mediates all the relation between X and Y. In other words, the impact of X on Y completely comes from its impact on Z




2.2. Z mediates part of the relation between X and Y. In other words, the impact of X on Y only partially comes from its impact on Z




What Are Moderator and Mediator? (Cont.)

3. The moderator changes the relation between X and Y. In other words, the relation between X and Y differ for different levels of the moderator. The moderator can occur before, after, or concurrently with the independent variable as long as it occurs before the occurrence of the outcome variable. In addition, the moderator can be the variable indicator the context where X and Y are embedded.




What Are Moderator and Mediator? (Cont.)

4. Moderated Mediation Model




What are Moderator and Mediator? (Cont.)

5. Mediated Moderation Model.



How to Estimate Mediation Effects?

Logic: If Z mediates the X-Y relation, then there should be a significant path linking X to Y through Z.




The logic of multiple regression is combining separate regression analyses to demonstrate the presence of the mediation effect

- X predicts Y
- X predicts Z
- When X and Z are both used to predict Y, but only Z remains a significant predictor of Y, such findings are interpreted as the bivariate relation between X and Y being totally mediated by Z


The logic of structural equation modeling (SEM) is to simultaneously estimate relations among X, Y, and Z and test if the indirect path (i.e., X-Z and Z-Y) is significant.

How to Estimate Moderation Effects?

Logic: If Z moderates the X-Y relation, the X-Y relation differs in magnitude or even sign for at least some levels of Z. For example:




(1) The X-Y relation for the whole sample




How to Estimate Moderation Effects? (Cont.)

(2) The strength of X-Y relation differs for subgroups of people



(3) The sign of the X-Y relation changes for subgroups of people



How to Estimate Moderation Effects? (Cont.)

The logic of regression: Adding an interaction term of X and Z in the regression and this interaction term indicates what additional effect X has on Y, depending on the level of Z.

$$Y = b_0 + aX + bZ + cXZ + \varepsilon$$

a indicates the main effect of X on Y

b indicates the main effect of Z on Y

c indicates the interaction effect of X and Z, i.e., the effect of X on Y, given the presence of Z

The logic of SEM:

Approach 1. Adding an interaction term of X and Z to the regression analysis.

Approach 2. Conduct separate regression analyses of Y and X for different levels of Z and then test if the regression coefficients are the same across these different regression analyses.

Stata Codes

```
*****  
* Read in data  
*****  
  
use "D:\jason\workshop\Moderating and Mediating\workshop.dta", clear  
des  
sum  
  
*****  
* 1. Mediation Analysis  
*****  
  
/*****  
1.1 Mediation within the regression framework  
  
1. test if X predict Y  
2. Test if X predict Z  
3. Test if X remains an important predictor when Z is used to predict Y  
*****/  
  
reg perform support  
reg satis support  
reg perform support satis  
  
/*****  
1.2 Mediation analysis within the structural equation modeling framework  
*****/  
  
sem (perform <- satis support) (satis <- support)  
  
estat teffects
```

```

/*****
1.3 Mediation analysis within the Generalized structural equation modeling framework
Use -nlcom- (Nonlinear combinations of estimators) to testimate the indirect and
total effects Nonlinear combinations of estimators
*****/

gsem (perform2 <- satis support) (satis <- support)

gsem, coeflegend

* obtain the direct effect of SUPPORT
nlcom _b[perform2:support]

* obtain indirect effect of SUPPORT
nlcom _b[perform2:satis]*_b[satis:support]

* obtain the total effect of SUPPORT
nlcom _b[perform2:support] +_b[perform2:satis]*_b[satis:support]

/*****
2. Moderation Analysis
*****/

/*****
2.1 OLS regression
*****/

reg perform c.support##i.group2
margins i.group2, at(support=(0.4(0.4)3.8))
marginsplot

```

```

/*****
2.2 Logistic Regression
*****/

logit perform2 c.support##i.group2

margins i.group2, at(support=(0.4(0.4)3.8))

marginsplot

/*****
2.3 Multiple group comparison within the SEM framework
*****/

sem ( perform <- support), group(group2) ginvariant(none)
estat ginvariant

/*****
2.4 Multiple group comparison within the GSEM framework
*****/

gsem ( perform <- support), group(group2)

gsem ( perform <- support), group(group2) ginvariant(none)

estimates store unconstrained

gsem ( perform <- support), group(group2) ginvariant(coef)
estimates store constrained

lrtest unconstrained constrained

```

```

/*****
3, Moderated mediation
*****/

/*****
3.1 Moderated Mediation within the SEM framework
*****/

sem (perform <- satis support) (satis <- support), group(group2)
estat teffects
estat ginvariant

/*****
3.2 Moderated Mediation within the GSEM framework
*****/

gsem (perform <- satis support) (satis <- support), group(group2) ginvariant(none)
estimates store unconstrained

gsem (perform <- satis support) (satis <- support), group(group2) ginvariant(coef)
|
estimates store constrained

lrtest unconstrained constrained

/*****
4, Mediated Moderation within the SEM framework
*****/

sem (perform <- satis support group2 group2_support) (satis <- support group2 group2_support)
estat teffects

```

Conclusions

- Moderation and mediation analysis can be viewed as two different ways to clarify the relation between X and Y. Moderation analysis examines whether the X-Y relation varies with the level of Z, while mediation examines whether the X-Y relation occurs through Z.
- Same variables can be conceptualized as moderators or mediators, depending on theories used. Thus, researchers need to consider what theories are used before conducting moderation and mediation analyses.
- When moderation analyses are conducted, Stata commands such as -margins-, and -marginsplot- are very useful for presenting the results. When mediation analyses are conducted, the use of path diagram can clearly present the conceptual model and the analysis results.
- Simple moderation and mediation can be examined with multiple regression and structural equation modeling. However, complex moderation and mediation models can only be examined by structural equation modeling.
- If you run into problems testing moderation and/or mediation, contact me at wuh@bgsu.edu